

**Math 2250 Maple Project 3: Numerical Methods
August 2008**

Due date: See the internet due dates. Maple lab 3 has four problems L3.1, L3.2, L3.3, L3.4.

References: Code in `maple` appears in `2250mapleL3-F2008.txt` at URL <http://www.math.utah.edu/~gustafso/>. This document: `2250mapleL3-F2008.pdf`. Other related and required documents are available at the web site:

- Numerical Solution of First Order DE (typeset, 19 pages, 220k pdf). A resource similar to the textbook, with maple examples and deeper detail. It is for a second reading, in case Edwards-Penney left too many questions unanswered.
- Sample Report for 2.4-3 (pdf 3 pages, 350k). This outline might be useful, if you are confused about which details to include.
- Numerical DE coding hints, TEXT Document (1 page, 2k). This document is appended here, for completeness. The web copy `2250mapleL3-F2008.txt` is suited for mouse copying.
- Sample maple code for Euler, Heun, RK4 (maple worksheet). Use `2250mapleL3-F2008-snips.mws` to load maple sample code without mouse copying.
- Sample maple code for exact/error reporting (maple worksheet). Normally not useful, because a hand calculator can do it faster.

Problem L3.1. (E & P Exercise 2.4-6, Symbolic Solution)

The symbolic solution of $y' = -2xy$, $y(0) = 2$ is $y = 2e^{-x^2}$. Using methods from the textbook, Chapter 1, display the details of the derivation for this symbolic solution, plus a full answer check.

Staple this page on top of your hand-written report

**Math 2250 Maple Project 3: Numerical Methods
F2008**

References: Code in `maple` appears in `2250mapleL3-F2008.txt` at URL <http://www.math.utah.edu/~gustafso/>. This document: `2250mapleL3-F2008.pdf`. Other related and required documents are available at the web site:

- Numerical Solution of First Order DE (typeset, 19 pages, 220k pdf). A resource similar to the textbook, with maple examples and deeper detail. It is for a second reading, in case Edwards-Penney left too many questions unanswered.
- Sample Report for 2.4-3 (pdf 3 pages, 350k). This outline might be useful, if you are confused about which details to include.
- Numerical DE coding hints, TEXT Document (1 pages, 2k). This document is appended here, for completeness. The web copy `2250mapleL3-F2008.txt` is suited for mouse copying.
- Sample maple code for Euler, Heun, RK4 (maple worksheet). Use `2250mapleL3-F2008-snips.mws` to load maple sample code without mouse copying.
- Sample maple code for exact/error reporting (maple worksheet). Normally not useful, because a hand calculator can do it faster.

Problem L3.2. (E & P Exercise 2.4-6)

Consider the initial value problem $y' = -2xy$, $y(0) = 2$ with symbolic solution $y = 2e^{-x^2}$. Apply Euler's method to produce two dot tables, as follows. The first has three rows, $h = 0.25$. The second has six rows, $h = 0.1$. Reproduce the table below and fill in missing digits. Follow the sample report for Edwards-Penney Exercise 2.4-3:

<http://www.math.utah.edu/~gustafso/2250SampleProblem2.4-3.pdf>

Reference L3.1 for the symbolic solution. Hand written work includes a check of the computer answer for $y(x_0 + h)$, that is, $y(0.25)$ for the program using $h = 0.25$ and $y(0.1)$ for the program using $h = 0.1$. The check makes sure the computer program obtains the correct answer for the very first step.

h	actual $y(.5)$	approx $y(.5)$
0.25	1.558	1.750?????
0.1	1.558	1.627?076?

Problem L3.3. (E & P Exercise 2.5-6)

Consider the initial value problem $y' = -2xy$, $y(0) = 2$ with symbolic solution $y = 2e^{-x^2}$. Apply Heun's method (Improved Euler) to produce one dot table of six rows, $h = 0.1$. Reproduce the table below and fill in missing digits. Follow the sample report for Exercise 2.4-3 as in problem L3.2 above. Hand written work includes a check of the computer answer for $y(x_0 + h)$, that is, $y(0.1)$ for the program using $h = 0.1$.

x	0.0	0.1	0.2	0.3	0.4	0.5
actual $y(x)$	2.000000000	1.980099667	1.921578878	1.827862371	1.704287578	1.557601566
approx $y(x)$	2.0000000	1.980????	1.921????	1.827????	1.704????	1.557????

Problem L3.4. (E & P Exercise 2.6-6)

Consider the initial value problem $y' = -2xy$, $y(0) = 2$ with symbolic solution $y = 2e^{-x^2}$. Apply the RK4 method to produce one dot table of three rows, $h = 0.25$. Reproduce the table below, filling in the missing digits. Follow the sample report for Exercise 2.4-3 as in problem L3.2 above. Forget hand calculator checks, because the table suggests that comparison with the symbolic solution is enough.

x	0.00	0.25	0.50
actual $y(x)$	2.000000000	1.878826126	1.557601566
approx $y(x)$	2.000000000	1.8788????0	1.557??329?

Staple this page on top of your hand-written and maple worksheet report

```

#Dr. Gustafson,
#I am in your 2250-4 class and am continually struggling with the codes for
#the problems in section 2.4, 2.5, and 2.6.  Actually 2.5 (Improved Euler)
#seems to be going pretty well, but I can't get very good results for 2.4
#(Euler) or 2.6 (Runge-Kutta Idea).  Is there a website where I may be able
#to find help and/or codes on these sections, as nothing I type in for my
#codes will seem to work?  Thanks for your time.

```

```

# =====
# Can't copy with the mouse? A work-around:
# =====
# Run the application "xclipboard &" to capture the mouse copies of this
# file.  Keep xclipboard near the xmaple window.  Go to the mozilla firefox
# window, copy with the mouse.  Switch to the xclipboard window.  Copy
# with the mouse from the xclipboard.  Then paste with mouse button 2 or
# mouse button 3 into xmaple.
# =====

```

```

# Maple code doesn't work? Read this:
# =====
# To type in a group, hold shift then press return, except
# for the last line of group, in which case use just return.
# If you copy multiple groups with the mouse, then split
# them using key F3 with the cursor placed at the front of
# a line where the split is to happen.

```

```

# Warning: These snips of code made for  $y'=1-x-y$ ,  $y(0)=3$ .
# Code computes approx values for  $y(0.1)$  to  $y(1.0)$ .
# 'Dots' is the list of dots for connect-the-dots graphics.
# =====

```

```

# Euler. Group 1, initialize.
f:=(x,y)->1-x-y;
x0:=0:y0:=3:h:=0.1:Dots:=[x0,y0]:n:=10;
# Group 2, repeat n times. Euler's method
for i from 1 to n do
  Y:=y0+h*f(x0,y0);
  x0:=x0+h:y0:=Y:Dots:=Dots,[x0,y0];
od;
# Group 3, display dots and plot.
Dots;
plot([Dots]);

```

```

# =====
# Heun. Group 1, initialize.
f:=(x,y)->1-x-y;
x0:=0:y0:=3:h:=0.1:Dots:=[x0,y0]:n:=10;
# Group 2, repeat n times. Heun method.
for i from 1 to n do
  Y1:=y0+h*f(x0,y0);
  Y:=y0+h*(f(x0,y0)+f(x0+h,Y1))/2;
  x0:=x0+h:y0:=Y:Dots:=Dots,[x0,y0];
od;
# Group 3, display dots and plot.
Dots;
plot([Dots]);

```

```

# =====
# RK4. Group 1, initialize.
f:=(x,y)->1-x-y;
x0:=0:y0:=3:h:=0.1:Dots:=[x0,y0]:n:=10;

```

```

# Group 2, repeat n times. RK4 method.
for i from 1 to n do
  k1:=h*f(x0,y0):
  k2:=h*f(x0+h/2,y0+k1/2):
  k3:=h*f(x0+h/2,y0+k2/2):
  k4:=h*f(x0+h,y0+k3):
  Y:=y0+(k1+2*k2+2*k3+k4)/6:
  x0:=x0+h:y0:=Y:Dots:=Dots,[x0,y0];
od:
# Group 3, display some dots and plot.
Dots[1],Dots[2],Dots[n+1];
plot([Dots]);

# Code snips for exact/error reports
# =====
# Making multiple curves on one plot
# =====
Exact:=x->2-x+exp(-x);          # An exact solution
plot({Exact(x),[Dots]},x=0..1/2); # plot exact and approx solutions

# =====
# How to create a Dots table for the exact solution
# =====
Exact:= x -> 2-x+exp(-x):n:=10:
ExactDots:=seq([Dots[j][1],Exact(Dots[j][1])],j=1..n+1);

# =====
# How to define and print percentage relative error:
# =====
P:=unapply(evalf(100*abs(exact-approx)/abs(exact)),(exact,approx));
ExactVal:=Exact(Dots[11][1]): # Compute exact y-value for x=1.0
ApproxVal:=Dots[11][2]:      # Get Euler approx y-value for x=1.0
P(ExactVal,ApproxVal);      # print percent relative error

# =====
# How to create a Dots table for percentage error
# =====
P:=unapply(evalf(100*abs(exact-approx)/abs(exact)),(exact,approx));
Pdots:=seq([Dots[j][1],P(Exact(Dots[j][1]),Dots[j][2])],j=1..11);

# =====
# Printing results and tables
# Make tables with a pencil, it saves time.
# =====
# To extract and print items 1,101,201,1001 from a list:
Dots1:=Dots[1],Dots[101],Dots[201],Dots[1001];

# =====
# Loop control
# =====
# To automate the production of a Dots list,
# enclose the desired code between 1 and 2 below.
# 1. for k from 1 to 10 do
# 2. od:
#
# Keyword "od:" is short for "end do:"
# Use ":" to stop loop results from printing.

# =====

```

```
# Debug
# =====
# To remove loop control and do it by hand, insert
# pound (#) signs as follows:
# 1. # for k from 1 to 10 do
# 2. # od:
# The hand-done loop is made by placing the mouse cursor
# in the group, then press return. Repeat for each loop step,
# which is 10 times for the loop above.
# the end
```

End of Maple Lab 3: Numerical Methods.